##  ITEM RESPONSE THEORY: TWO-PARAMETER LOGISTIC MODEL
##  MODEL FOR MEASURMENT OF POLITICAL KNOWLEDGE

model {
	# LOOP OVER N RESPONDENTS
	for (i in 1:N) {        
		# LOOP OVER K ITEMS
	    for (k in 1:K) {    
	    	# LOGISTIC MODEL FOR POLITICAL KNOWLEDGE
	    	y[i,k] ~ dbern (p[i,k])
	    	logit(p[i,k]) <- beta[k]*(theta[i] - alpha[k])
	    }    
	    # DISTRIBUTIONAL ASSUMPTION FOR THE LATENT TRAIT
	    theta[i] ~ dnorm (0, 1)
    }    
             
	# DISTRIBUTIONS OF ITEM PARAMETERS
    for (k in 1:K) {
    	beta[k] ~ dnorm (1, 0.5) T(0,)  # DISCRIMINATION PARAMETER; USUALLY [0.5, 3]
    	alpha[k] ~ dnorm (0, tau.alpha)  # DIFFICULTY PARAMETER
    }   

	tau.alpha ~ dgamma (0.01, 0.01) # shape and rate;
	sigma.alpha <- 1/sqrt(tau.alpha)

} # END OF MODEL


